
Introduction 
 

Everyone knows that Excel works brilliantly with numbers, but not everyone is aware that Excel can 

work just as well with dates and to a lesser extent, with times.  

In fact Excel treats dates just like numbers. Day 1 in Excel is 1/1/1900.  

While days are treated as whole numbers, times are treated as fractions. Hence, 1.5 is noon on 

1/1/1900.   

You have probably seen what happens when the format is 

removed from a date cell (Clear Formats) or you copy and use 

paste special values from a date cell into another cell. 

 

 

 

 

 

 

It displays a number above 40,000. This number represents how 

many days since 1/1/1900. 

 

We will examine dates first and then have a look at how you work with times in Excel. 

When working with dates formatting can be an issue. 

When referring to months in column headings, always use the 1st of the month.    

Date Calculations 
 

Since dates are treated like numbers doing calculations based on dates is reasonably straight 

forward. You need to be aware that Excel will apply a default format to cells it thinks are dates. That 

isn’t always what you want, or expect. 

Default Date Formats 

 

In this example if you have two dates and subtract one 

date from the other Excel will automatically default 

the format of the calculation cell (C2) to a date format.  

 

In this case that doesn’t help and you can either 

remove the format using Clear Formats icon or apply 

a number format to correct it.  

The previous example shows that subtraction 

calculations are exclusive. In the example subtracting the first day of the month from the last day of 

the month gives you one day less than the number of days in the month. It is basically 28 - 1 = 27. 

There is a formula to calculate the number of days in a given month and it will be covered later in 

the session.     

Add days to a date 

A common requirement is to enter a date, say the end of the 

week, and then use a formula to calculate the end of subsequent 

weeks.  

That is as easy as adding 7 to the date. 

Cell E3 has been copied down to create a list of Sunday 

weekending dates. 

 

If you need to add weekdays or workdays to a date you can use the WORKDAYS function. The 

WORKDAYS function allows you to specify a start date and add a number of weekdays to it to arrive 

at a completion date. You can include a list of holidays to be excluded the weekdays.    

WORKDAYS(Start_Date,Days,Holidays) 

Start_Date  - usually a cell reference containing a date 

Days - the number of workdays to add to the Start_Date 

Holidays - (optional) - is a range containing a list of dates to be treated as public holidays and to be 

excluded from the calculation 

The formula in cell C2 is 

=WORKDAY(A2,B2) 

This has been copied down the column. This formula adds weekdays to the start date. No public 

holidays are taken into account. 

The formula in cell D2 is 

=WORKDAY(A2,B2,$F$2:$F$4) 

This has been copied down the column. This formula adds weekdays to the start date but the dates 

in F2:F4 are treated as public holidays and not worked. 

There are no public holidays in February so there is no difference between C2 and D2, but there are 

differences between rows 3 and 4 because there are public holiday dates in the date ranges 

involved. 

Adding months to Date 

 

We saw how easy adding a week to a date is. Adding a month is more difficult because of our 

different length months. You can use the EDATE function to easily add a number of months to a 

date.  

EDATE(Start_Date,Months) 

Start_Date - usually a cell containing the start date  

Months - the number of months to add to the date, usually a keyed in value. May be 1 to add a 

month or 3 for quarters. 

 

The formula in cell C2 is 

=EDATE(A2,B2)  

This has been copied down the column. 

As you can see it correctly handles the 

end of month issues for February in rows 

5, 6 and 7.   

First and Last day of the Month 

 

Two common dates that you need to calculate are the first and last days of the month. 

There are two functions that both handle these two requirements. The EOMONTH function handles 

end of month calculations and can be manipulated to arrive at the first of the month. 

The EOMONTH function returns the last day of the month for the month it refers to. You can add a 

number of months to a date and return the last day of the month for the calculated month.   

EOMONTH(Date,Months) 

Date – usually a cell containing a date 

Months – the number of months to add to the date. 

The DATE function allows you to build a date based on three inputs, for year, month and day. 

DATE(Year,Month,Day) 

Year – a number representing the year of the date 

Month - a number representing the month of the date. If a number greater than 12 is used it will 

adjust the  month – eg 13 will return January of the next year referred to in the year argument. 

Day - a number representing the day of the date. If a number greater than the last day of the month 

it will adjust the  month – eg 32 in January will refer to 1st February. Large values can also increment 

the year. 

The formula in cell D2 is 

=EOMONTH(A2,B2) 

This is straightforward as it takes the start date and adds a number of months and return the month 

end date for that calculated month. 

The formula in cell C2 is  

=EOMONTH(A2,B2-1)+1 

This is less straightforward. It uses the fact that the EOMONTH returns the last day of the month and 

removes 1 from the added month (inside the brackets) and then adds one to the result (outside the 

brackets) which in effect adds one to the last day of the previous month to arrive at the first day of 

the current month. (You may need to read that a few times.) 

The formula in cell E2 is 

=DATE(YEAR(A2),MONTH(A2)+B2,1) 

This uses the YEAR and MONTH functions (they return their respective values from a date) to build a 

date with the day value being hard-keyed as 1. 

The formula in cell F2 is 

=DATE(YEAR(A2),MONTH(A2)+B2+1,0) 

Note the use of a zero day. This takes advantage of the way the day value works in the DATE 

function. One is added to the month calculation and then 0 is used for the day. Zero is one less than 

one and hence it adjusts the date back to the last day of the previous month, which is why one was 

added to the month. 

Today’s Date 

 

Sometimes you need to do calculations based on today’s date 

eg has a due date expired.  

You can uses today’s date in a formula by using the function 

TODAY(). This provides the current day’s date. This is dynamic, 

it uses your computer’s date. If you need to return the date 

and time use the NOW() function which include the fraction for 

the time. 

    

The formula in cell B2 is 

=IF(A2<TODAY(),"Yes","") 

This formula has been copied down. 

Based on the date that this manual was created the list is 

accurate. 

Compare it to the report on the day of the webinar.  

 

 

 

If you want to enter today’s date in a cell so that is 

doesn’t change then use the keyboard shortcut Ctrl + 

; this is not dynamic, it is a data entry and won’t 

change.  

Current time is Ctrl+ Shift + : 

Try using the mouse to right-drag a date and see the 

Fill Weekdays option - handy for rosters.  

Days of the week 

If you need to work with Monday, Tuesday etc there are a couple of ways to do it. You can convert 

the date into its day equivalent using a format or the TEXT function. You can also use the WEEKDAY 

function. 

Weekdays - format 

To identify what day of the 

week a date is, you can use 

a custom number format.  

In the Custom Category of 

the Number tab of Format 

Cells you can use 

dddd 

to display the full day 

name or 

 

ddd  

to display the short day name. 

Column B has a formula that links to column A. This means the date value is still in the cell, but it is 

displayed as the day of the week. 

 

Weekdays using a TEXT function 

Cell C2 has the formula 

=TEXT(A2,"dddd") 

This returns the text as shown to the cell 

- this is no longer a date, but text.  

 

The WEEKDAY Function 

 

The WEEKDAY 

function returns 

a number from 1 

to 7.  

 

1 = Sunday, 2 = Monday through to 7 = Saturday. 

  



Excel Date and Time Calculations by Neale Blackwood  A4 Accounting 

Page 9 of 15 
© 2017 Neale Blackwood 

Calculation using the WEEKDAY function 

The SUMPRODUCT function, which is Excel’s most versatile function, can use other functions within 

its brackets to perform different types of summing.    

The formula in cell F2 is 

=SUMPRODUCT((WEEKDAY($A$2:$A$21)=D2)*($B$2:$B$21)) 

 

The formula has been copied down. 

This formula summarises the listing in 

columns A and B into total weekday 

sales. 

 

 

 

 

 

 

 

 

 

 

 

 

TEXT Function and Dates 

If you need to create automated headings or narrations then you may have struck this issue with 

Excel. 

The formula in cell 

C2 is  

=A2&" "&B2 

The & symbol joins 

text together. 

 

You can see that the number of the date has been displayed rather than the words.  

To create an automated heading you can use the TEXT function to convert the date to text.   

The formula in D2 is 

=A2&" "&TEXT(B2,"mmmm yy") 

The format which is the second part of the TEXT function uses the same type of formats seen in the 

Custom Number format earlier. You must enclose them in quotation marks. 

You can also use this technique to format $ values in narrations or sentences. 

 

Workday Calculation between two dates 

Another type of workday calculation is you have a start date and an end date and you need to 

calculate how many workdays in between. The NETWORKDAYS function handles this type of 

calculation.   

NETWORKDAYS(Start_Date,End_Date,Holidays) 

Start_Date and End_Date are usually cell references. 

Holidays - is a range containing dates to be treated as public holidays and not worked. 

  

 

The formula in cell C2 is 

=NETWORKDAYS(A2,B2) 

This calculates weekdays between the two dates, as it has no holiday range. 

The formula in cells D2 is 

=NETWORKDAYS(A2,B2,$F$2:$F$5) 

This calculates workdays as it includes a list of holidays to exclude from the calculations.  

There is also a NETWORKDAYS.INTL (Excel 2010 and later versions) that allows you to specify 

weekends other than Saturday and Sunday. 

Time Calculations 
As mentioned earlier time is treated as a fraction of a day.  

0.25  = 6am 

0.5 = noon 

0.75 = 6pm 

Time entry 

The main issue you have with time is whether to treat is as a decimal, as above, or as per the normal 

digital time display eg 6:15 or 18:45. 

To enter time in a cell you need to use the colon to 

separate the hours and the minutes.  

 

 

If the minutes are zero, then you can just enter the hours 

followed by the colon. 

  

 

 

If you type 11:1 and press Enter you are in fact entering 

one minute past 11 not 10 minutes passed.  

You must include the zero at the end for minutes. 

 

  

 

 

 

Once the hours are entered then you can subtract one 

from the other to get the difference in hours. 

Once you have hours in a cell 

there are some limitations to 

using them in calculations.  

If you multiply by hours you will 

find this issue. 

 

Hours are still a fraction of a 

day.  

 

Hence, to do any multiplications you need to multiply the hours by 24 to convert them to a decimal 

value. 

Another problem arises when the time different goes 

over a day. 

Excel fills the cell with # signs when it can’t perform the 

time calculation. 

 

The adjusted formula in cell 

C2 that handles going over a 

day is 

=IF(B2>=A2,B2-A2,B2+1-A2) 

 

The solution involves adding a whole day (1) to the end time to make it more than the start time and 

then the subtraction works. 

Adding up time 

 

Another problem you can face with time is 

that Excel won’t let you display more the 24 

hours. 

To get around this limitation you need to 

use a Custom Number format to force Excel 

to display hours larger than 24.   

 

 



Excel Date and Time Calculations by Neale Blackwood  A4 Accounting 

Page 13 of 15 
© 2017 Neale Blackwood 

 

The custom number format to use is 

[h]:mm 

Other Date Calculations 
If you need to calculate how many years or months between two dates you can use the DATEDIF 

function. This function is unusual in that it doesn’t appear in Excel’s help system and there is no help 

to use it as you create it. 

Warning: Apparently this function can be unstable in some circumstances. 

DATEDIF(Start_Date,End_Date,Type) 

Start_Date and End_Date are usually cell references. 

Type - is a text string in quotation marks that specifies the type of calculation to perform. See table 

below. 

Type Description 

y Full years between dates 

m Full months between dates 

ym Remainder months after years  

md Remainder days after months  

yd Remainder days after years 

 

Examples 

 

Total number of completed years between dates - the formula in cell C2 is  

=DATEDIF(A2,B2,"y") 

Total number of completed months between dates - the formula in cell D2 is  

=DATEDIF(A2,B2,"m") 

Total remaining months after completed years - the formula in cell E2 is  

=DATEDIF(A2,B2,"ym") 

Total remaining completed days after completed months - the formula in cell F2 is  

=DATEDIF(A2,B2,"md") 

Common Date Calculations 
 

Automated Date Ranges 

 

In financial models and some reports you typically have a central date range that specifies the dates 

to be used in all the sheets in the model and report. 

One way to automate this is to use the EDATE function. You have a starting date and then add a 

month to it or 3 months for a quarter. 

 

Formula in C1 is  

=EDATE(B1,1) 

Formula in C3 is  

=EDATE(B3,3) 

Days in the month 

If B1 contains a date the following formula will calculate how many calendar days in the month. 

=DAY(DATE(YEAR(B1),MONTH(B1)+1,0)) 

 

Removing Times from Dates  

Sometimes when you import dates into Excel they may include a time fraction. The time is typically 

redundant. To remove a time from a date use the TRUNC function (short for truncate) which will 

remove the decimal values leaving just the date as a whole number. It doesn’t do any rounding. 

 

Random Dates 

If you need to generate random dates 

for testing purposes, you can use the 

RANDBETWEEN function. It requires a 

starting number (date) and an ending 

number (date) – the result must be 

formatted to display properly. It is a 

whole number – no fraction.      


